
Одиниця у квадраті
Умова
Число, що складається з n (1 ≤ n ≤ 654321) одиниць, підносять до квадрату. Знайти результат такої операції.
	Приклад

	INPUT.TXT
	5

	OUTPUT.TXT
	123454321


Розв’язок

Якщо число містить n цифр, то нескладно перевірити, що його квадрат міститиме близько 2·n цифр, отже результат, який ми шукаємо, не вміщується у статичну пам’ять, якою може розпоряджатися Паскаль.
Давайте для початку розв’яжемо задачу "в лоб". Будемо множити два числа з n одиниць за правилом довгої арифметики – в стовпчик. При цьому ми можемо отримувати цифри результату в зворотньому порядку (з кінця), а оскільки нам не можна усі цифри зберігати, будемо записувати їх у файл.
В результаті одержимо те, що треба, але у зворотньому порядку. Тепер приступимо до інверсії вихідного файлу (перевернення результату).
Таку операцію неможливо виконати з текстовим файлом в Pascal (тут можливий лише послідовний доступ) без використання додаткового файлу, а останнє заборонено правилами будь-якої олімпіади. Проте це можливо, якщо файл, який ми отримали, відкрити, як типізований файл (з прямим доступом до даних) file of char. Тут Pascal дає нам карти у руки – ми можемо обміняти перший символ з останнім, другий – з передостаннім і так далі.
program squared_N_ones;

var N,i,j,p,size:longint;

    fi,fo:text;

    f:file of char;

    a,b:char;

begin
  Assign (fi,'input.txt'); Reset (fi);

  ReadLn (fi,N);           Close (fi);

  Assign (fo,'output.txt'); Rewrite (fo);

  p:=0;

  for i:=1 to N do
      begin
        Write (fo,(i+p) mod 10);

        p:=(i+p) div 10;

      end;

  for i:=N-1 downto 1 do
      begin
        Write (fo,(i+p) mod 10);

        p:=(i+p) div 10;

      end;

  Close (fo);

  Assign (f,'output.txt'); Reset (f);

  size:=FileSize (f);

  for i:=1 to size div 2 do
      begin
        Seek (f,i-1);    Read (f,a);

        Seek (f,size-i); Read (f,b);

        Seek (f,i-1);    Write (f,b);

        Seek (f,size-i); Write (f,a);

      end;

  a:=#10; b:=#13; Seek (f,size-1); Write (f,a,b);

  Close (f);

end.
Після такого "лобового" підходу давайте все ж пошукаємо закономірність, за якою можна було б відновити вигляд шуканого числа за відомими даними.
На перший погляд здасться, що шукане число матиме вигляд послідовних цифр 123 ... n ... 321, проте ця закономірність виконується лише для n ≤ 9, а що ж для інших чисел?
Розглянемо детальніше наш "квадрат" на прикладі n=15:

                111111111111111
                111111111111111
                111111111111111
               1111111111111110
              11111111111111100
             111111111111111000
            1111111111111110000
           11111111111111100000
          111111111111111000000
         1111111111111110000000
        11111111111111100000000
       111111111111111000000000
      1111111111111110000000000
     11111111111111100000000000
    111111111111111000000000000
   1111111111111110000000000000
  11111111111111100000000000000
  12345679012345654320987654321   - результат
Для отримання результату ми спочатку додаємо одиницю та n-1 нулів в останньому з кінця розряді, потім дві одиниці у другому... в десятому з кінця розряді ми додаємо десять одиниць, записуємо 0 в результат і 1 в перенос. Потім, відповідно, до наступних 11 одиниць додається ще одна – з переносу, тобто в цифрах результату утворюється "зсув" (після 0 іде не 1, а 2).
Коли ж кількість одиниць перестає збільшуватись (в нашому прикладі це 15-та цифра з кінця), то відповідно і цифри послідовно зменшуються, а зсув відбувається, коли черговий перенос зменшується на одиницю, тобто коли чергова цифра – 9.
Таким чином, загальний вигляд результату такий:



Тут блок спадання повторюється стільки разів, скільки одиниць переносу накопичилося при додаванні. Скільки ж їх? Одиниця переносу з’являється, як було вже сказано, коли цифра результату дорівнює 0, тобто через кожні дев’ять цифр. Отож одиниць переносу буде n div 9. Відповідно, блоків зростання у нас стільки ж. Запис числа завершується дописуванням цифри 1, про яку ми забули при такій класифікації частин відповіді.
Між блоками спадання і зростання, в самій середині числа, відбувається "перелом". Переломною цифрою буде та, що стоїть під першою цифрою співмножників у записі дії множення "у стовпчик", тобто n-та з кінця. Блок цифр від 1 "вгору" до переломної цифри і від неї "вниз" до 2 не належить до перерахованих блоків. Його необхідно генерувати окремо.
Таким чином, в процесі виконання описаного способу генерації відповіді ми можемо виводити у вихідний файл згенеровані цифри одразу, не зберігаючи їх у пам’яті.
Описаний алгоритм реалізовується дуже просто:

program squared_N_ones;

const DG:array [1..9] of byte = (1,2,3,4,5,6,7,9,0);

      IG:array [1..9] of byte = (0,9,8,7,6,5,4,3,2);

var fi,fo:text;

    N,i,j:integer;

begin
  Assign (fi,'input.txt'); Reset (fi);
  ReadLn (fi,N);           Close (fi);
  Assign (fo,'output.txt'); Rewrite (fo);

  if N mod 9=0 then
     begin
       for i:=1 to N div 9-1 do
           for j:=1 to 9 do Write (fo,DG[j]);

       for i:=1 to 9 do Write (fo,i);

       for i:=8 downto 2 do Write (fo,i);

       for i:=1 to N div 9-1 do
           for j:=1 to 9 do Write (fo,IG[j]);

     end else

     begin

       for i:=1 to N div 9 do
           for j:=1 to 9 do Write (fo,DG[j]);

       if N mod 9<>1 then
          begin

            for i:=1 to N mod 9 do Write (fo,i);

            for i:=N mod 9-1 downto 2 do Write (fo,i);

          end;

       for i:=1 to N div 9 do
           for j:=1 to 9 do Write (fo,IG[j]);

     end;

  WriteLn (fo,1);

  Close (fo);

end.
Зверніть увагу на граничні випадки, розглянуті у програмі.

По-перше, коли n mod 9=0, то блоків переносу буде (n div 9-1), оскільки "переломний" блок складається з елементів 1…9…2 (тобто 9 + 8 = 17 елементів, які не повинні бути враховані ні у блоках зростання, ні у блоках спадання). По-друге, коли n mod 9=1, то переломний блок взагалі буде відсутнім, адже остання одиничка переносу, яка мала б виникнути при множенні, "поглинається" тим, що одиниць, які додаються у наступному розряді, вже на одну менше. Фактично, ця одиничка з блоку зростання переходить у блок спадання, який її "нейтралізує".
Такі дрібні, але фатальні помилки, як неврахування граничних випадків, часто зустрічаються навіть у професійних програмістів, тому вони потребують особливої уваги.
Прості пари
Умова

У класі вчаться n (n≤500000) учнів. Вчитель математики вирішив  розсадити їх за парти таким чином, щоб сума номерів у списку в класному журналі кожних двох сусідів по парті була простим числом.

Розподіліть їх на найбільшу кількість таких пар довільним одним з можливих способів.
	Приклад

	INPUT.TXT
	7

	OUTPUT.TXT
	1 6
7 4

5 2


Розв’язок
Запропонуємо один з можливих підходів до розв’язання даної задачі.
Давайте знайдемо перше просте число більше за N. Позначимо його через P. Тоді пари чисел (P-N, N), (P-N+1, N-1) … мають сумарну вагу P і задовольняють умові задачі. Легко помітити, що усі числа від P-N до N входять в одну з перерахованих пар.

Після цього залишилося розв’язати ту ж задачу для нового значення N, а саме (P-N-1). Цей процесс продовжується до тих пір, поки значення N більше 1. При цьому ми одержимо максимальну кількість пар різних чисел від 1 до N, які задовольняють умові задачі.
progam Prime_pairs;

var n,p,i,j:longint;

    fi,fo:text;

function prime(N:longint):boolean;

var i:longint;

begin
  i:=2;

  while (i*i<=N)and(N mod i>0) do Inc (i);

  prime:=(N in [2,3])or(i*i>N);

end;

begin
  Assign (fi,'input.txt'); Reset (fi); Read (fi,n); Close (fi);

  Assign (fo,'output.txt'); Rewrite (fo);

  while N>1 do
        begin
          p:=(N+1) div 2*2+1;

          while not prime(p) do Inc (p,2);

          i:=p-N; j:=N; N:=i-1;

          while i<j do
                begin
                  WriteLn (fo,i,' ',j);

                  Inc (i); Dec (j);

                end;

        end;

  Close (fo);

end. 

Ви бачите, що при розв’язанні ми користувались тим же принципом зведення задачі до своєї ж підзадачі, що у попередньому випуску – прицип динамічного програмування.

Зауважимо, що для заданих в умові обмежень не має великого значення, яким алгоритмом перевірки простоти числа користуватися, тому ми використали найпростіший з них. Про деякі інші алгоритми буде детально розказано у наступному випуску.
Підсумок
Сьогодні ми розібрали ще одну методику роботи з великими об’ємами даних. Це використання файлу з прямим доступом. Початково типізовані файли були розроблені для роботи з базами даних: додавання запису, перегляд, коректування даних. Проте вони розв’язують руки програмісту у випадку, коли йому необхідно працювати з неосяжними об’ємами даних. Файл з прямим доступом є замінником оперативної пам’яті, він широко використовується у сучасних операційних системах (так званий "файл підкачки"). Проте не слід забувати, що звертання до файлу значно уповільнює роботу програми. У всьому є свої мінуси...
1234567901234.....1...(х-1)х(х-1)....2098765432098...0987654321

